```
Quit[];
```

In[1]:=

(*SETUP*)

```
(*CONFIGURATION VARIABLES*)
q = {{x[t]}, {y[t]}, {θ1[t]}, {θsb[t]}};


(*TRANSFORMATIONS*)
(*Divers body in the world frame*)
gWB = {{Cos[θ1[t]], -Sin[θ1[t]], 0, x[t]},
   {Sin[θ1[t]], Cos[θ1[t]], 0, y[t]}, {0, 0, 1, 0}, {0, 0, 0, 1}};

(*Divers feet frame in world frame*)
gWF = gWB.{{1, 0, 0, 0}, {0, 1, 0, -1/2}, {0, 0, 1, 0}, {0, 0, 0, 1}};

(*Spring board wall contact frame in world frame*)
gWSbp = {{Cos[(π/2) - θsb[t]], Sin[(π/2) - θsb[t]], 0, 0},
   {-Sin[(π/2) - θsb[t]], Cos[(π/2) - θsb[t]], 0, 0}, {0, 0, 1, 0}, {0, 0, 0, 1}};

(*Spring board center of mass frame in world frame*)
gWSb = gWSbp.{{1, 0, 0, lsb/2}, {0, 1, 0, wsb/2}, {0, 0, 1, 0}, {0, 0, 0, 1}};

(*Spring contact point with the board in world frame*)
gWSpc = gWSb.{{1, 0, 0, (lsb/2) - .3}, {0, 1, 0, (-wsb/2)}, {0, 0, 1, 0}, {0, 0, 0, 1}};

(*Divers contact point with the board in world frame*)
gWDcp = gWSb.{{1, 0, 0, (lsb/2) - .2}, {0, 1, 0, (wsb/2)}, {0, 0, 1, 0}, {0, 0, 0, 1}};



(*USEFULL FUNCTIONS*)
(*HAT AND UNHAT OPERATIONS FOR so3*)
VecToso3[ω_] :=
  {{0, -ω[[3, 1]], ω[[2, 1]]}, {ω[[3, 1]], 0, -ω[[1, 1]]}, {-ω[[2, 1]], ω[[1, 1]], 0}};

(*HAT AND UNHAT OPERATIONS FOR SE3*)
VecTose3[V_] :=
  ArrayFlatten[{{VecToso3[{V[[1 ;; 3, 1]]}ᵀ], {V[[4 ;; 6, 1]]}ᵀ}, {0, 0}}];

se3ToVec[Vskew_] := {{Vskew[[3, 2]], Vskew[[1, 3]],
     Vskew[[2, 1]], Vskew[[1, 4]], Vskew[[2, 4]], Vskew[[3, 4]]}}ᵀ;

(*TRANS INVERSE*)
TransInv[T_] :=
  ArrayFlatten[
   {{T[[1 ;; 3, 1 ;; 3]]ᵀ, -T[[1 ;; 3, 1 ;; 3]]ᵀ.{T[[1 ;; 3, 4]]}ᵀ}, {0, 1}}];
```

```
(*Piecewise solutions*)
xsol = Piecewise[{{0, t > 0 && t < 0}}];
ysol = Piecewise[{{0, t > 0 && t < 0}}];
θ1sol = Piecewise[{{0, t > 0 && t < 0}}];
θsbsol = Piecewise[{{0, t > 0 && t < 0}}];

(*PARAMETERS*)
g = 9.8;
l = 1;
lsb = 2;
m = 1;
msb = 1;
J = 1;
w = 0.4;
wsb = 0.2;
k = 1000;(*Spring Constant*)

springlength = √(((1.7 Sin[π / 2]) - 1.7) ^2 + ((-1.7 Cos[π / 2]) - (-1.2)) ^2);
Δspring = springlength - √(((gWSpc[[1, 4]]) - 1.4) ^2 + ((gWSpc[[2, 4]]) - (-1.2)) ^2);

(*LAGRANGIAN*)
IMMatrix = {{m, 0, 0, 0, 0, 0}, {0, m, 0, 0, 0, 0},
   {0, 0, m, 0, 0, 0}, {0, 0, 0, J, 0, 0}, {0, 0, 0, 0, J, 0}, {0, 0, 0, 0, 0, J}};

IMMatrixSB = {{msb, 0, 0, 0, 0, 0}, {0, msb, 0, 0, 0, 0},
   {0, 0, msb, 0, 0, 0}, {0, 0, 0, J, 0, 0}, {0, 0, 0, 0, J, 0}, {0, 0, 0, 0, 0, J}};


(*Kinetic Energy*)
VbDiverBody = se3ToVec[((TransInv[gWB]).(D[gWB, t]))];
KEDiverBody = (1 / 2) * (VbDiverBodyᵀ.IMMatrix.VbDiverBody);

VbSpringBoard = se3ToVec[((TransInv[gWSb]).(D[gWSb, t]))];
KESpringBoard = (1 / 2) * (VbSpringBoardᵀ.IMMatrixSB.VbSpringBoard);
(*Potential Energy*)
VDiverBody = m * g * gWB[[2, 4]];
VSpringBoard = (1 / 2) * k * (Δspring) ^2 + msb * g * gWSb[[2, 4]];
(*Spring potential plus gravity*)


L = KEDiverBody + KESpringBoard - VDiverBody - VSpringBoard;



(*Before
```

Impact*)

```
(*EQUATIONS*)
Eq1 = D[D[L, x'[t]], t] - D[L, x[t]] == 0;
Eq2 = D[D[L, y'[t]], t] - D[L, y[t]] == 0;
Eq3 = D[D[L, θ1'[t]], t] - D[L, θ1[t]] == 0;
Eq4 = D[D[L, θsb'[t]], t] - D[L, θsb[t]] == 0;


(*Solve for x'' y'' θ1'' θ2'' θ3''*)
ELtemp = Solve[Eq1 && Eq2 && Eq3 && Eq4, {x''[t], y''[t], θ1''[t], θsb''[t]}];

(*Write Euler Lagrange equations in a list
 format because this is how NDSolve wants them!!*)
EL = {x''[t] == ELtemp[[1, 1, 2]], y''[t] == ELtemp[[1, 2, 2]],
   θ1''[t] == ELtemp[[1, 3, 2]], θsb''[t] == ELtemp[[1, 4, 2]]};

(*INITIAL CONDITIONS*)
InitCon = {x'[0] == 1, x[0] == 0.8, y'[0] == 0,
   y[0] == 2.5, θ1'[0] == 1, θ1[0] == 0, θsb[0] == (1), θsb'[0] == 0};

(*Solve the DE*)(*The integration is stopped when the
 feet of the diver equalt the y of point on the spring board*)
sol = NDSolve[Join[EL, InitCon], {x[t], y[t], θ1[t], θsb[t]},
   {t, 0, 10}, Method → {"EventLocator",
     "Event" :→ -1/2 Cos[θ1[t]] + 1.8` Cos[θsb[t]] - 0.2` Sin[θsb[t]] + y[t],
     "EventAction" :→ Throw[tend1 = t, "StopIntegration"]}];
Print ["The impact is at time ", tend1]

(*Save Solutions to piecewise functions*)
xsol = Piecewise[{{sol[[1, 1, 2]], t > 0 && t ≤ tend1}}];
ysol = Piecewise[{{sol[[1, 2, 2]], t > 0 && t ≤ tend1}}];
θ1sol = Piecewise[{{sol[[1, 3, 2]], t > 0 && t ≤ tend1}}];
θsbsol = Piecewise[{{sol[[1, 4, 2]], t > 0 && t ≤ tend1}}];
```

(*During

Impact*)

```
(*CONSTRAINT*)
ϕ1 = (-1/2 Cos[θ1[t]] + 1.8` Cos[θsb[t]] - 0.2` Sin[θsb[t]] + y[t]);
(*Bottom hits contact point*)


p1 = D[L, x'[t]];
p2 = D[L, y'[t]];
p3 = D[L, θ1'[t]];
p4 = D[L, θsb'[t]];

H = p1[[1]] * x′[t] + p2[[1]] * y'[t] + p3[[1]] * θ1'[t] + p4[[1]] * θsb'[t] - L;


(*IMPACT UPDATE EQUATIONS*)
Hτp =
  H /. sol /. {x'[t] → xτp, y'[t] → yτp, θ1'[t] → θ1τp, θsb'[t] → θsbτp} /. t → tend1;
Hτm = H /. sol /. {x'[t] → D[x[t] /. sol, t], y'[t] → D[y[t] /. sol, t],
     θ1'[t] → D[θ1[t] /. sol, t], θsb'[t] → D[θsb[t] /. sol, t]} /. t → tend1;
EQ1 = (Flatten[Hτp] - Flatten[Hτm]) [[1]] == 0;
EQ2 =
  Flatten[p1 /. sol /. {x'[t] → xτp, y'[t] → yτp, θ1'[t] → θ1τp, θsb'[t] → θsbτp} /. t →
          tend1] [[1]] -
      Flatten[p1 /. sol /. {x'[t] → D[x[t] /. sol, t], y'[t] → D[y[t] /. sol, t],
          θ1'[t] → D[θ1[t] /. sol, t], θsb'[t] → D[θsb[t] /. sol, t]} /.
        t → tend1] [[1]] == λ * D[ϕ1, x[t]] /. sol /. t → tend1;
EQ3 = Flatten[p2 /. sol /. {x'[t] → xτp, y'[t] → yτp, θ1'[t] → θ1τp, θsb'[t] → θsbτp} /.
        t → tend1] [[1]] -
      Flatten[p2 /. sol /. {x'[t] → D[x[t] /. sol, t], y'[t] → D[y[t] /. sol, t],
          θ1'[t] → D[θ1[t] /. sol, t], θsb'[t] → D[θsb[t] /. sol, t]} /.
        t → tend1] [[1]] == λ * D[ϕ1, y[t]] /. sol /. t → tend1;
EQ4 = Flatten[p3 /. sol /. {x'[t] → xτp, y'[t] → yτp, θ1'[t] → θ1τp, θsb'[t] → θsbτp} /.
        t → tend1] [[1]] -
      Flatten[p3 /. sol /. {x'[t] → D[x[t] /. sol, t], y'[t] → D[y[t] /. sol, t],
          θ1'[t] → D[θ1[t] /. sol, t], θsb'[t] → D[θsb[t] /. sol, t]} /.
        t → tend1] [[1]] == λ * D[ϕ1, θ1[t]] /. sol /. t → tend1;
EQ5 = Flatten[p4 /. sol /. {x'[t] → xτp, y'[t] → yτp, θ1'[t] → θ1τp, θsb'[t] → θsbτp} /.
        t → tend1] [[1]] -
      Flatten[p4 /. sol /. {x'[t] → D[x[t] /. sol, t], y'[t] → D[y[t] /. sol, t],
          θ1'[t] → D[θ1[t] /. sol, t], θsb'[t] → D[θsb[t] /. sol, t]} /.
        t → tend1] [[1]] == λ * D[ϕ1, θsb[t]] /. sol /. t → tend1;
EQ6 = λ ≠ 0;
```

```
NewInitialConditions =
  NSolve[{EQ1[[1]], EQ2[[1]], EQ3[[1]], EQ4[[1]], EQ5[[1]], EQ6},
   {xτp, yτp, θ1τp, θsbτp, λ}];

(*NEW INIT CONDITIONS*)
xp = x[t] /. sol /. t → tend1;
xp = xp[[1]];
xv = NewInitialConditions[[1, 1, 2]];

yp = y[t] /. sol /. t → tend1;
yp = yp[[1]];
yv = NewInitialConditions[[1, 2, 2]];

θ1p = θ1[t] /. sol /. t → tend1;
θ1p = θ1p[[1]];
θ1v = NewInitialConditions[[1, 3, 2]];

θsbp = θsb[t] /. sol /. t → tend1;
θsbp = θsbp[[1]];
θsbv = NewInitialConditions[[1, 4, 2]];

(*EQUATIONS*)
Eq21 = D[D[L, x'[t]], t] - D[L, x[t]] == 0;
Eq22 = D[D[L, y'[t]], t] - D[L, y[t]] == 0;
Eq23 = D[D[L, θ1'[t]], t] - D[L, θ1[t]] == 0;
Eq24 = D[D[L, θsb'[t]], t] - D[L, θsb[t]] == 0;


(*Solve for x'' y'' θ1'' θsb''*)
ELtemp2 = Solve[Eq21 && Eq22 && Eq23 && Eq24, {x''[t], y''[t], θ1''[t], θsb''[t]}];
EL2 = {x''[t] == ELtemp2[[1, 1, 2]], y''[t] == ELtemp2[[1, 2, 2]],
   θ1''[t] == ELtemp2[[1, 3, 2]], θsb''[t] == ELtemp2[[1, 4, 2]]};

(*NEW INITIAL CONDITIONS*)
NewInitCon = {x'[tend1] == xv, x[tend1] == xp, y'[tend1] == yv, y[tend1] == yp,
   θ1'[tend1] == θ1v, θ1[tend1] == θ1p, θsb[tend1] == θsbp, θsb'[tend1] == θsbv - 1};

(*Solve the DE*)
sol2 = NDSolve[Join[EL2, NewInitCon], {x[t], y[t], θ1[t], θsb[t]}, {t, tend1, 10}];


(*Save Solutions to piecewise functions*)
xsol = Piecewise[{{xsol, 0 ≤ t ≤ tend1}, {sol2[[1, 1, 2]], t > tend1 && t ≤ 10}}];
ysol = Piecewise[{{ysol, 0 ≤ t ≤ tend1}, {sol2[[1, 2, 2]], t > tend1 && t ≤ 10}}];
θ1sol = Piecewise[{{θ1sol, 0 ≤ t ≤ tend1}, {sol2[[1, 3, 2]], t > tend1 && t ≤ 10}}];
θsbsol = Piecewise[{{θsbsol, 0 ≤ t ≤ tend1}, {sol2[[1, 4, 2]], t > tend1 && t ≤ 10}}];
```

```
(*Expanding the piecewise to animate*)
xExpand = PiecewiseExpand[xsol];
yExpand = PiecewiseExpand[ysol];
θ1Expand = PiecewiseExpand[θ1sol];
θsbExpand = PiecewiseExpand[θsbsol];

(*PLOTS*)
(*Plot[{xExpand,yExpand},{t,0,10}, AxesLabel→{t,{x,y}}]
 Plot[{θ1Expand,θsbExpand},{t,0,10}, AxesLabel→{t,{θ1,θsb}}]*)

(*Animation*)
(*BODY*)
bod1 = {-w / 2, -l / 2, 0, 1};
bod1t[T_] := (((gWB.bod1) /. x[t] → xExpand /. y[t] → yExpand /. θ1[t] → θ1Expand /.
        θsb[t] → θsbExpand) /. t → T)[[1 ;; 2]];
bod2 = {w / 2, -l / 2, 0, 1};
bod2t[T_] := (((gWB.bod2) /. x[t] → xExpand /. y[t] → yExpand /. θ1[t] → θ1Expand /.
        θsb[t] → θsbExpand) /. t → T)[[1 ;; 2]];
bod3 = {w / 2, l / 2, 0, 1};
bod3t[T_] := (((gWB.bod3) /. x[t] → xExpand /. y[t] → yExpand /. θ1[t] → θ1Expand /.
        θsb[t] → θsbExpand) /. t → T)[[1 ;; 2]];
bod4 = {-w / 2, l / 2, 0, 1};
bod4t[T_] := (((gWB.bod4) /. x[t] → xExpand /. y[t] → yExpand /. θ1[t] → θ1Expand /.
        θsb[t] → θsbExpand) /. t → T)[[1 ;; 2]];

(*Spring Board*)
Sb1 = {-lsb / 2, -wsb / 2, 0, 1};
Sb1t[T_] :=
  (((gWSb.Sb1) /. x[t] → xExpand /. y[t] → yExpand /. θ1[t] → θ1Expand /.
        θsb[t] → θsbExpand) /. t → T) [[1 ;; 2]];
Sb2 = {lsb / 2, -wsb / 2, 0, 1};
Sb2t[T_] :=
  (((gWSb.Sb2) /. x[t] → xExpand /. y[t] → yExpand /. θ1[t] → θ1Expand /.
        θsb[t] → θsbExpand) /. t → T) [[1 ;; 2]];
Sb3 = {lsb / 2, wsb / 2, 0, 1};
Sb3t[T_] :=
  (((gWSb.Sb3) /. x[t] → xExpand /. y[t] → yExpand /. θ1[t] → θ1Expand /.
        θsb[t] → θsbExpand) /. t → T) [[1 ;; 2]];
Sb4 = {-lsb / 2, wsb / 2, 0, 1};
Sb4t[T_] :=
  (((gWSb.Sb4) /. x[t] → xExpand /. y[t] → yExpand /. θ1[t] → θ1Expand /.
        θsb[t] → θsbExpand) /. t → T) [[1 ;; 2]];

(*Spring contact point*)
SpCPt[T_] :=
  (((gWSpc.{0, 0, 0, 1}) /. x[t] → xExpand /. y[t] → yExpand /. θ1[t] → θ1Expand /.
```

```
        θsb[t] → θsbExpand) /. t → T)[[1 ;; 2]];

(*Spring*)
Spring2D[start_, end_, loops_, radius_] :=
 Module[{detail = 40, steps}, steps = detail (loops + .5);
  Translate[Rotate[Line@Table[
      {radius + (Norm[end - start] - 2 radius) a / steps + radius Cos[2 Pi a / detail + Pi],
       radius Sin[2 Pi a / detail]}, {a, 0, steps}], {{1, 0}, end - start}], start]]


Movie = Animate[Graphics
   [{Black, Thick, Spring2D[{1.4, -1.2}, SpCPt[t], 5, 0.15],
     Line[{{{0, -1.2}, {lsb, -1.2}}}], (*{1.87819,-1.38729}*)
     Line[{{{0, -2}, {0, 8}}}],
     Line[{bod1t[t], bod2t[t]}],
     Line[{bod2t[t], bod3t[t]}],
     Line[{bod3t[t], bod4t[t]}],
     Line[{bod4t[t], bod1t[t]}],
     Line[{Sb1t[t], Sb2t[t]}],
     Line[{Sb2t[t], Sb3t[t]}],
     Line[{Sb3t[t], Sb4t[t]}],
     Line[{Sb4t[t], Sb1t[t]}]},
    PlotRange → {{-0.5, 5}, {-2, 4}}],
   {t, 0.001, 2.5, .001}, AnimationRunning → False]

The impact is at time 0.439002
```

Out[110]=